Costruire archi di parabola

Per utilizzare una curva nel linguaggio GDL al fine di costruire oggetti, conviene scriverla in forma parametrica. Come parametro si utilizza un ipotetico valore temporale `t`, immaginando che si tratti dell'equazione del moto di un punto materiale che si muova da `Po=(xo,yo)` per `t=0` fino a `Pf=(xo,yo)` per `t=1`. Con questo metodo si possono costruire sia curve piane in 2D che sghembe in 3D.

Nel caso della parabola conviene servirsi della formula di Bézier di grado 2, a 3 nodi, dove il nodo intermedio `Pi` è il punto di incontro delle tangenti in `Po`, `Pf`.

In termini vettoriali l'equazione è:

```
P=Po*(1-t)^2+2*Pi*(1-t)*t+ Pf*t^2
```

che in 2D corrisponde a:

```
x=xo*(1-t)^2+2*xi*(1-t)*t+ xf*t^2
y=yo*(1-t)^2+2*yi*(1-t)*t+ yf*t^2
```

mentre in 3D basta aggiungere la terza coordinata:

```
z=zo*(1-t)^2+2*zi*(1-t)*t+ zf*^2
```

Per chiarire il significato geometrico dei nodi poniamo:

```
Po=(-lu/2,0),  Pi=(0,2*h),  Pf=(lu/2,0)
```

La formula diventa:

```
x=-lu/2*(1-t)^2 +lu/2*^2=lu*t-lu /2
y=4*h*(1-t)*t=4*h*(t-t^2)
```



Osservando il risultato si deduce che la proiezione su x si muove di moto uniforme a partire da `x=-lu/2` per giungere a `x=lu/2` in `t=1`, percorrendo uno spazio pari a `lu`.
Lungo l'asse `y` il moto è uniformemente accelerato a partire da `y=0`, con culmine in `t=1/2` dove risulta: `y(1/2)=h`
Dunque `h` è l'altezza dell'arco e `lu` ne è la luce.

Se trasliamo `Pi` della lunghezza `xi` parallelamente all'asse `x: Pi=(xi,2*h),` la parabola si inclina. Di fatto si applica una affinità, per cui si inclina pure l'asse della parabola, che non coinciderà più con il punto di massimo a quota `h`.

Nel caso generale conviene porre: `Pm=(Po+Pf)/2`
ovvero `Pm` è il punto medio del vettore `Po_Pf`:

```
xm=(xo+xf)/2
ym=(yo+yf)/2
```

1. Arco di parabola definito dagli estremi e dal punto intermedio

Se `Pv=(xv,yv)` è il vertice relativo dell'arco (che coincide con quello effettivo della parabola solo se il vettore `Pm_Pv` è ortogonale a `Po_Pf`), questo si troverà a metà strada tra `Pm` e `Pi`:

```
Pv=Pm+Pm_Pi/2
xv=xm+(xi-xm)/2=xi/2+xm/2=xi/2+(xo+xf)/4
yv= yi/2+(yo+yf)/4
xi=2*xv-(xo+xf)/2
yi=2*yv-(yo+yf)/2
```

L'equazione generale dell'arco di parabola potrà allora essere scritta come :

```
x=xo*(1-t)^2+(4*xv-(xo+xf))*(1-t)*t+ xf*t^2
y=yo*(1-t)^2+(4*yv-(yo+yf))*(1-t)*t+ yf*t^2
```

equivalente a :

```
x=xo*(1+2*t^2-3*t)+4*xv*(t-t^2)+xf*(2*t^2-t)
y=yo*(1+2*t^2-3*t)+4*yv*(t-t^2)+yf*(2*t^2-t)
```

In questo caso il nodo di incontro delle tangenti risulta essere `Pi=Pm+2*Pm_Pv` ovvero :

```
xi=xm+2*(xv-xm)= 2*xv-xm
yi=ym+2*(yv-ym)= 2*yv-ym
```

L'angolo `ano` rispetto l'orizzontale della tangente in `Po` è quello del vettore `To=Po_Pi`

```
tox=xi-xo=2*xv-xo/2-xf/2-xo= 2*xv-3/2*xo-xf/2
toy= 2*yv-3/2*yo-yf/2
```

Per calcolare l'angolo di un vettore occorre applicare una formula un po' complessa che restituisce un angolo entro `[0,360)`, cioè con `0` compreso e `360` escluso.
In questo caso risulta:

```
mo= sqr(tox^2+toy^2)
ano=acs(tox/mo)*(1+sgn(toy)-abs(sgn(toy)))+180*( abs(sgn(toy))-sgn(toy))
```

Analogamente l'angolo `anf` rispetto l'orizzontale della tangente in `Pf` è quello del vettore

```
Tf=Pf_Pi
tfx= 2*xv-3/2*xf-xo/2
tfy= 2*yv-3/2*yf-yo/2
mf= sqr(tfx^2+tfy^2)
anf=acs(tfx/mf)*(1+sgn(tfy)-abs(sgn(tfy)))+180*( abs(sgn(tfy))-sgn(tfy)
```

Per tradurre le formule negli algoritmi del GDL occorre prima definire i parametri:

```
n=       !numero intero corrispondente ai vertici sulla curva
xo=      !punto iniziale
yo=
xv=      !massimo dell'arco
yv=
xf=      !punto finale
yf=

for i=0 to n
t=i/n
x=xo*(1+2*t^2-3*t)+4*xv*(t-t^2)+xf*(2*t^2-t)
y=yo*(1+2*t^2-3*t)+4*yv*(t-t^2)+yf*(2*t^2-t)
put x,y
next i
```

Se ci troviamo nel 2D possiamo richiamare i valori immagazzinati in memoria sfruttando ad esempio il comando `poly2` :

```
poly2 nsp/2 ,1 , get (nsp)
```

Per completare l'arco tracciamo anche le tangenti e l'asse relativo:

```
xm=(xo+xf)/2 :ym=(yo+yf)/2
xi= 2*xv-xm  :yi= 2*yv-ym
line2 xo,yo, xi,yi: line2 xf,yf, xi,yi: line2 xv,yv, xi,yi
line2 (xo+xi)/2, (yo+yi)/2, (xf+xi)/2, (yf+yi)/2
```

Il quarto segmento è la tangente al punto di massimo dell'arco. Volendo anche tracciare gli angoli relativi alle tangenti, prima li determiniamo riferiti a `x+`, ovvero all'asse delle ascisse:

```
tox= 2*xv-3/2*xo-xf/2
toy= 2*yv-3/2*yo-yf/2
mo= sqr(tox^2+toy^2)
ano=acs(tox/mo)*(1+sgn(toy)-abs(sgn(toy)))+180*( abs(sgn(toy))-sgn(toy))

tfx= 2*xv-3/2*xf-xo/2
tfy= 2*yv-3/2*yf-yo/2
mf= sqr(tfx^2+tfy^2)
anf=acs(tfx/mf)*(1+sgn(tfy)-abs(sgn(tfy)))+180*( abs(sgn(tfy))-sgn(tfy))
```

Per riferire tali angoli al vettore `Po_Pf` invece che a `x+` occorre calcolarne l'angolo relativo:

```
xof= xf-xo
yof= yf-yo
md = sqr( xof^2+ yof^2)
aof=acs(xof/md)*(1+sgn(yof)-abs(sgn(yof)))+180*( abs(sgn(yof))-sgn(yof))

arc2 xo,yo, md/7, 0, aof
line2 xo,yo, xo+md/7,yo

arc2 xo,yo, md/5, aof, ano
line2 xo,yo, xo+md/5*cos(aof),yo+md/5* sin(aof)

afo=acs(-xof/md)*(1-sgn(yof)-abs(sgn(yof)))+180*( abs(sgn(yof))+sgn(yof))
arc2 xf,yf, md/5, anf, afo
line2 xf,yf, xf+md/5*cos(afo),yf+md/5* sin(afo)
```

Per l'angolo su `Pf` mi sono riferito al vettore opposto a `Po_Pf`, che forma rispetto `x+` l'angolo `afo` supplementare all'angolo `aof`.

2. *Arco di parabola definita dagli estremi e dalle inclinazioni delle tangenti agli estremi*

Se invece si preferisce dare per noti i valori angolari delle tangenti, da cui dedurre il nodo `Pi`, conviene riferirli alla retta individuata da `Po_Pf`. Se tali angoli li indichiamo con `ato` e `atf`, detto `md` il modulo di `Po_Pf`, vale la formula:

```
md = sqr( (xf-xo)^2+ (yf-yo)^2)
2*h=md * tan(ato)*tan(atf) / (tan(atf)-tan(ato))=
md / ( cos(ato)/sin(ato)- cos(atf)/sin(atf) )
d = 2*h/tan(atf) = 2*h*cos(atf)/sin(atf)
```

Dove `h` è la semidistanza del nodo `Pi` da `Po_Pf`, mentre `d` è la distanza con segno da `Pf` della proiezione ortogonale di `Pi` su `Po_Pf` da `Pf`.
Dovrà risultare `sin(atf)><0, sin(ato)><0, atf><ato`



Il nodo `Pi` è individuato dalla somma dei vettori `Pf`, `Pf_Q` e `Q_Pi` :

```
Pf_Q=d*(cos(aof),sin(aof)) , Q_Pi=2*h*(cos(aof+90),sin(aof+90))
xi=xf+ 2*h*cos(atf)/sin(atf)*cos(aof)-2*h*sin(aof)
yi=yf+ 2*h*cos(atf)/sin(atf)*sin(aof)+2*h*cos(aof)
```

In conclusione, dati gli angoli delle tangenti riferiti alla retta orientata individuata dal vettore `Po_Pf`, la formula diventa:

```
xo=
yo=
xf=
yf=
ato=
atf=
xof= xf-xo : yof= yf-yo : md = sqr( xof^2+ yof^2)
aof=acs(xof/md)*(1+sgn(yof)-abs(sgn(yof)))+180*( abs(sgn(yof))-sgn(yof))

if not(abs(sgn(sin(ato)))=0 or abs(sgn(sin(atf)))=0 or abs(sgn(atf-ato))=0) then

h=md / ( cos(ato)/sin(ato)- cos(atf)/sin(atf) )/2
xi= xf+ 2*h*(cos(atf)/sin(atf)*cos(aof)-sin(aof))
yi= yf+ 2*h*(cos(atf)/sin(atf)*sin(aof)+cos(aof))
x=xo*(1-t)^2+2*xi*(1-t)*t+ xf*t^2
y=yo*(1-t)^2+2*yi*(1-t)*t+ yf*t^2

endif
```

Nelle figure sottostanti sono rappresentati archi costruiti con questo metodo.